

Maxima workbook for Principles of NMR Spectroscopy

Chapter 11: THE MATHEMATICAL FORMALISM OF QUANTUM MECHANICS

1 Introduction

This wxMaxima workbook is an electronic supplement to to the book Principles of NMR Spectroscopy: An Illustrated Guide, David P. Goldenberg, University Science Books, (c) 2016. This and related files are available for download through links at: <http://uscibooks.com/goldenberg.htm> wxMaxima is a graphical user interface to the computer algebra system (CAS) Maxima. General information about Maxima and wxMaxima, along with free versions of the programs, can be found at: <http://maxima.sourceforge.net/> and <http://andrejv.github.io/wxmaxima/> Before attempting to use this workbook, users are strongly encouraged to read and experiment with the introductory workbook, gettingStarted.wxmx This software is distributed under the conditions of the BSD license and without any guarantees or warranties. (c) 2015 by David P. Goldenberg Please send comments, including bug reports, to this address:
David P. Goldenberg
Department of Biology
University of Utah
257 South 1400 East
Salt Lake City, UT 84112-0840
goldenberg@biology.utah.edu

Chapter 11 covers, in addition to a general introduction to the formalism of quantum mechanics, the treatment of isolated spin-1/2 particles, including the wavefunction and magnetization operators. This workbook includes the various calculations presented in that chapter and can be used to carry out additional calculations.

This workbook uses definitions in 1spinLib.mac for quantum mechanics of a single spin 1/2 particle. The calculations are carried out using matrix representations of the wavefunctions and operators, as explained in Appendix E. The 1spinLib.mac file can be stored in any location accessible to the user, but it is recommended that it and the accompanying file, 2spinLib.mac, be stored in the .maxima subdirectory of the users home directory. This facilitate loading the macro files without specifying a directory path. Before, using the definitions in 1spinLib.mac, however, the next section introduces the representation of wavefunctions and operators as matrices and vectors, starting from scratch.

2 Wavefunctions and operators represented as vectors and matrices.

The wavefunction for an arbitrary, isolated spin-1/2 particle can be represented as a superposition of two mutually orthogonal wavefunctions. For many purposes, the most convenient basis functions are the eigenfunctions if the Iz-magnetization operator, which we will write here in Dirac form as |alpha> and |beta>. A superposition state is written as psi = ca*|alpha> + cb*|beta> where ca and cb are complex-valued coefficients. In the vector representation, only the coefficients are explicitly written, and they are placed in a column vector. To ensure that ca and cb are recognized as complex variables, we explicitly declare them as such

```
(%i1) declare(ca,complex);
```

```
(%o1)  done
```

```
(%i2) declare(cb,complex);
```

```
(%o2)  done
```

We then define the variable k_psi as the 2x1 matrix, or column vector

```
(%i3) k_psi:matrix([ca],[cb]);
```

```
(%o3)   $\begin{pmatrix} ca \\ cb \end{pmatrix}$ 
```

```
(%i4) k_psi;
```

```
(%o4)   $\begin{pmatrix} ca \\ cb \end{pmatrix}$ 
```

Note that this matrix is distinct from both the list containing ca and cb, [ca,cb] and the 1x2 matrix that would be specified by matrix([ca,cb])

(%i5) [ca,cb];

(%o5) [ca,cb]

(%i6) matrix([ca,cb]);

(%o6) (ca cb)

The column vector represents a ket in Dirac notation, and the bra is represented by the corresponding row vector containing the complex conjugates of the coefficients.

(%i7) b_psi:transpose(conjugate(k_psi));

(%o7) (ca cb)

In versions 15.04 and later (presumably) of wxMaxima, complex conjugates are represented by the overstrike

Operators are represented as 2x2 matrices, and the action of the operator is represented as a matrix multiplication. For the Iz operator, the matrix is:

(%i8) Iz:matrix([1/2,0],[0,-1/2]);

(%o8) (1/2 0; 0 -1/2)

(%i9) Iz.k_psi;

(%o9) (ca/2; -cb/2)

To calculate the average outcome of a measurement of Iz for an arbitrary wavefunction, we multiply the product Iz.k_psi by b_psi, being careful with the order of operation

(%i10) b_psi.Iz.k_psi;

(%o10) (ca*ca-bar)/2 - (cb*cb-bar)/2

We can use Maxima to find the normalized eigenfunctions of Iz, expressed as vectors. (See the examples in gettingStarted.wmx for more details) But, first, we must load the eigen package.

(%i11) load(eigen);

(%o11) /usr/local/Cellar/maxima/5.37.2/share/maxima/5.37.2/share/matrix/eigen.mac

(%i12) uniteeigenvectors(Iz);

(%o12) [[[-1/2, 1/2], [1, 1]], [[0, 1], [1, 0]]]

This result indicates that the eigen values are -1/2 and 1/2. The eigenvectors are represented in the output as lists: [0,1] and [1,0] Although the vectors can be used in this form, it is better and safer to explicitly define them as column vectors.

(%i13) k_a:matrix([1],[0]);

(%o13) (1; 0)

(%i14) k_b:matrix([0],[1]);

(%o14) (0; 1)

The bras corresponding to the kets are

```
(%i15) b_a:transpose(conjugate(k_a));
```

```
(%o15)  (1  0)
```

```
(%i16) b_b:transpose(conjugate(k_b));
```

```
(%o16)  (0  1)
```

We can then demonstrate that these vectors do, indeed, represent the eigenfunctions of the I_z operator

```
(%i17) k_a;
```

```
(%o17)  (1)  
        (0)
```

```
(%i18) Iz.k_a;
```

```
(%o18)  (1/2)  
        (0)
```

```
(%i19) k_b;
```

```
(%o19)  (0)  
        (1)
```

```
(%i20) Iz.k_b;
```

```
(%o20)  (0)  
        (-1/2)
```

We can also demonstrate that the eigenfunctions of I_z are orthonormal.

```
(%i21) b_a.k_a;
```

```
(%o21)  1
```

```
(%i22) b_b.k_b;
```

```
(%o22)  1
```

```
(%i23) b_a.k_b;
```

```
(%o23)  0
```

```
(%i24) b_b.k_a;
```

```
(%o24)  0
```

The definitions shown above, along with others, are included in the 1spinLib.mac macro file. At this point, we will clear all of the definitions and start over with the calculations in Chapter 11, using the macro file.

```
(%i25) kill(values);
```

```
(%o25)  done
```

Then, we load the macro file. If this file has been placed in the users .maxima directory, it can be loaded without explicitly indicating the path

```
(%i26) load("1spinLib.mac")$
```

All of the defined variables and functions can be listed with the values and functions commands, respectively.

```
(%i27) values;
```

```
(%o27)  [hermitianmatrix, nondiagonalizable, knowneigvals, knowneigvects, listeigvects, listeigvals, rightmatrix, leftmatrix, k_psi, k_a, k_b, b_a, b_b]
```

(%i28) functions;

(%o28) [innerproduct(x,y), unitvector(x), columnvector(x), gramschmidt(x, [myinnerproduct]), eigenvalues(mat), eigenvectors(mat), su

Some of the variables and functions listed above are defined by the Maxima packages eigen and nchrpl, which are loaded by 1spinLib.mac

The assigned values of the variables can be displayed by simply evaluating the variables

(%i29) k_a;

(%o29) $\begin{pmatrix} 1 \\ 0 \end{pmatrix}$

(%i30) Ix;

(%o30) $\begin{pmatrix} 0 & \frac{1}{2} \\ \frac{1}{2} & 0 \end{pmatrix}$

The definition of functions is displayed using the fundef function, which can be accessed from the "Maxima" menu of wxMaxima. For instance ket(a,b) is a command for simplifying, slightly, the construction of a column vector from the complex coefficients.

(%i31) fundef(ket);

(%o31) ket(a,b) := columnvector([a,b])

bra(k) creates a bra from an existing ket

(%i32) fundef(bra);

(%o32) bra(k) := transpose(\bar{k})

Some of the variables and functions defined 1spinLib.mac are described and used below, following the presentation and calculations in Chapter 11. Others will be introduced in subsequent chapters.

Chapter 11

We begin here to follow the treatment of the spin-1/2 system in Section 11.4, repeating some of the material above

1 11.4 The spin-1/2 system

1.1 11.4.1 The Iz operator

Page 328

The eigenfunctions of Iz, represented as vectors are:

(%i33) k_a;

(%o33) $\begin{pmatrix} 1 \\ 0 \end{pmatrix}$

(%i34) k_b;

(%o34) $\begin{pmatrix} 0 \\ 1 \end{pmatrix}$

(%i35) Iz.k_a;

(%o35)

$$\begin{pmatrix} \frac{1}{2} \\ 0 \end{pmatrix}$$

(%i36)

$$I_z.k_b;$$

(%o36)

$$\begin{pmatrix} 0 \\ -\frac{1}{2} \end{pmatrix}$$

The bras of the eigenfunctions are

(%i37)

$$b_a;$$

(%o37)

$$\begin{pmatrix} 1 & 0 \end{pmatrix}$$

(%i38)

$$b_b;$$

(%o38)

$$\begin{pmatrix} 0 & 1 \end{pmatrix}$$

The Four Rules

Rule 1. The eigenvectors have unit length. Page 328

(%i39)

$$b_a.k_a;$$

(%o39)

$$1$$

(%i40)

$$b_b.k_b;$$

(%o40)

$$1$$

2. The eigenvectors are orthogonal Page 329

(%i41)

$$b_a.k_b;$$

(%o41)

$$0$$

(%i42)

$$b_b.k_a;$$

(%o42)

$$0$$

Rule 3. Any wavefunction for a spin-1/2 particle can be written as a linear combination of |alpha> and |beta> Page 32

In vector representation, the general form of a wavefunction for a single spin is:

(%i43)

$$k_{psi};$$

(%o43)

$$\begin{pmatrix} ca \\ cb \end{pmatrix}$$

The general form of the complex conjugate, bra, of the wavefunction is:

(%i44)

$$b_{psi};$$

(%o44)

$$\begin{pmatrix} \overline{ca} & \overline{cb} \end{pmatrix}$$

The product of the bra and ket

(%i45)

$$b_{psi}.k_{psi};$$

(%o45)

$$cb \cdot \overline{cb} + ca \cdot \overline{ca}$$

The operation of Iz on the general wavefunction

(%i46)

$$I_z.k_{psi};$$

$$(\%o46) \quad \begin{pmatrix} \frac{ca}{2} \\ -\frac{cb}{2} \end{pmatrix}$$

Calculating the average outcome of a measurement of Iz on an arbitrary wavefunction

```
(%i47) b_psi.Iz.k_psi;
```

$$(\%o47) \quad \frac{ca \cdot \overline{ca}}{2} - \frac{cb \cdot \overline{cb}}{2}$$

The 1spinLib.mac library includes a function for calculating the average outcome from an operator and wavefunction. The function takes two arguments, the operator and the wavefunction

```
(%i48) meanPsi(Iz,k_psi);
```

$$(\%o48) \quad -\frac{cb \cdot \overline{cb} - ca \cdot \overline{ca}}{2}$$

Rule 4. The probability equation Page 331
The probability of a measurement, applied to a wavefunction psi yielding a specific eigenvalue, lambda_i, corresponding to eigenfunction psi_i is given by
P_i = |<psi_i|psi>|^2

The probability that an arbitrary wavefunction will give rise to the value 1/2, associated with eigenfunction |alpha>, when Iz is measured

```
(%i49) abs(b_a.k_psi)^2;
```

$$(\%o49) \quad |ca|^2$$

The probability that an arbitrary wavefunction will give rise to the value -1/2, associated with eigenfunction |beta>, when Iz is measured

```
(%i50) prob(b_b,k_psi);
```

$$(\%o50) \quad |cb|^2$$

1.2 11.4.2 Other angular momentum operators

Page 332

The Ix operator The matrix form is:

```
(%i51) Ix;
```

$$(\%o51) \quad \begin{pmatrix} 0 & \frac{1}{2} \\ \frac{1}{2} & 0 \end{pmatrix}$$

When applied to an arbitrary wavefunction

```
(%i52) Ix.k_psi;
```

$$(\%o52) \quad \begin{pmatrix} \frac{cb}{2} \\ \frac{ca}{2} \end{pmatrix}$$

As shown above, Maxima can find the normalized eigenvectors and eigenvalues for a matrix.

```
(%i53) uniteigenvectors(Ix);
```

$$(\%o53) \quad [[[-\frac{1}{2}, \frac{1}{2}], [1, 1]], [[[\frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}]], [[\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}]]]]$$

The eigenvector and uniteigenvector functions output the vectors in the form of lists. In some contexts, it is preferable to form proper column vectors. The 1spinLib.mac file includes a function for forming kets from the values of the complex coefficients. We will name the eigenvectors for Ix k_elx1 and k_elx2


```
(%i54) k_eIx1:ket(1/sqrt(2),1/sqrt(2));
```

(%o54)
$$\begin{pmatrix} \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} \end{pmatrix}$$

```
(%i55) k_eIx2:ket(1/sqrt(2),-1/sqrt(2));
```

(%o55)
$$\begin{pmatrix} \frac{1}{\sqrt{2}} \\ -\frac{1}{\sqrt{2}} \end{pmatrix}$$

The second eigenvector shown above is not exactly the same as the one shown in the book, but they are related by a factor of -1. This does not actually change any measurable results, but we will redefine it for consistency with the book.

```
(%i56) k_eIx2:ket(-1/sqrt(2),1/sqrt(2));
```

(%o56)
$$\begin{pmatrix} -\frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} \end{pmatrix}$$

Another 1spinLib.mac function creates a bra from a ket

```
(%i57) fundef(bra);
```

(%o57)
$$\text{bra}(k) := \text{transpose} \left(\overline{k} \right)$$

```
(%i58) b_eIx1:bra(k_eIx1);
```

(%o58)
$$\left(\frac{1}{\sqrt{2}} \quad \frac{1}{\sqrt{2}} \right)$$

```
(%i59) b_eIx2:bra(k_eIx2);
```

(%o59)
$$\left(-\frac{1}{\sqrt{2}} \quad \frac{1}{\sqrt{2}} \right)$$

To show that they are, indeed, normalized, we multiply each ket by its corresponding bra:

```
(%i60) b_eIx1.k_eIx1;
```

(%o60) 1

```
(%i61) b_eIx2.k_eIx2;
```

(%o61) 1

To show that they are orthogonal:

```
(%i62) b_eIx1.k_eIx2;
```

(%o62) 0

```
(%i63) b_eIx2.k_eIx1;
```

(%o63) 0

To show that they are eigenvectors of Ix

```
(%i64) Ix.k_eIx1;
```

(%o64)
$$\begin{pmatrix} \frac{1}{2^{\frac{3}{2}}} \\ \frac{1}{2^{\frac{3}{2}}} \end{pmatrix}$$

In Maxima, we can divide the elements of one matrix by the corresponding elements of another using the ordinary division sign. We can use this to show that the result of the previous operation is, indeed, a simple multiple of k_elx1

(%i65) %/k_eIx1;

(%o65)
$$\begin{pmatrix} \frac{1}{2} \\ \frac{1}{2} \end{pmatrix}$$

Similarly for the other eigenvector

(%i66) Ix.k_eIx2;

(%o66)
$$\begin{pmatrix} \frac{1}{2^{\frac{3}{2}}} \\ -\frac{1}{2^{\frac{3}{2}}} \end{pmatrix}$$

(%i67) %/k_eIx2;

(%o67)
$$\begin{pmatrix} -\frac{1}{2} \\ -\frac{1}{2} \end{pmatrix}$$

We can calculate the average Ix magnetization for |alpha> and |beta>, or a general wavefunction

(%i68) b_a.Ix.k_a;

(%o68) 0

(%i69) b_b.Ix.k_b;

(%o69) 0

(%i70) b_psi.Ix.k_psi;

(%o70)
$$\frac{ca \cdot \overline{cb}}{2} + \frac{\overline{ca} \cdot cb}{2}$$

(%i71) meanPsi(Ix,k_psi);

(%o71)
$$\frac{\overline{ca} \cdot cb + ca \cdot \overline{cb}}{2}$$

The Iy operator

(%i72) Iy;

(%o72)
$$\begin{pmatrix} 0 & -\frac{i}{2} \\ \frac{i}{2} & 0 \end{pmatrix}$$

(%i73) Iy.k_psi;

(%o73)
$$\begin{pmatrix} -\frac{i \cdot cb}{2} \\ \frac{i \cdot ca}{2} \end{pmatrix}$$

The eigenvectors are found as before

(%i74) uniteigenvectors(Iy);

(%o74)
$$[[[-\frac{1}{2}, \frac{1}{2}], [1, 1]], [[[\frac{1}{\sqrt{2}}, -\frac{i}{\sqrt{2}}]], [[\frac{1}{\sqrt{2}}, \frac{i}{\sqrt{2}}]]]]$$

Expressed as column vectors, the eigenvectors are:

(%i75) k_eIy1:ket(1/sqrt(2),%i/sqrt(2));

$$(\%o75) \quad \begin{pmatrix} \frac{1}{\sqrt{2}} \\ \frac{i}{\sqrt{2}} \end{pmatrix}$$

$$(\%i76) \text{ k_eIy2:ket}(1/\text{sqrt}(2),-\%i/\text{sqrt}(2));$$

$$(\%o76) \quad \begin{pmatrix} \frac{1}{\sqrt{2}} \\ -\frac{i}{\sqrt{2}} \end{pmatrix}$$

The corresponding bras are:

$$(\%i77) \text{ b_eIy1:bra}(k_eIy1);$$

$$(\%o77) \quad \left(\begin{array}{cc} \frac{1}{\sqrt{2}} & -\frac{i}{\sqrt{2}} \end{array} \right)$$

$$(\%i78) \text{ b_eIy2:bra}(k_eIy2);$$

$$(\%o78) \quad \left(\begin{array}{cc} \frac{1}{\sqrt{2}} & \frac{i}{\sqrt{2}} \end{array} \right)$$

Demonstrating that they are eigenvectors of ly

$$(\%i79) \text{ Iy.k_eIy1};$$

$$(\%o79) \quad \begin{pmatrix} \frac{1}{2^{\frac{3}{2}}} \\ \frac{i}{2^{\frac{3}{2}}} \end{pmatrix}$$

$$(\%i80) \text{ \%}/k_eIy1;$$

$$(\%o80) \quad \begin{pmatrix} \frac{1}{2} \\ \frac{1}{2} \end{pmatrix}$$

$$(\%i81) \text{ Iy.k_eIy2};$$

$$(\%o81) \quad \begin{pmatrix} -\frac{1}{2^{\frac{3}{2}}} \\ \frac{i}{2^{\frac{3}{2}}} \end{pmatrix}$$

$$(\%i82) \text{ \%}/k_eIy2;$$

$$(\%o82) \quad \begin{pmatrix} -\frac{1}{2} \\ -\frac{1}{2} \end{pmatrix}$$

The eigenfunctions are orthonormal

$$(\%i83) \text{ b_eIy1.k_eIy1};$$

$$(\%o83) \quad 1$$

$$(\%i84) \text{ b_eIy2.k_eIy2};$$

$$(\%o84) \quad 1$$

$$(\%i85) \text{ b_eIy1.k_eIy2};$$

$$(\%o85) \quad 0$$

$$(\%i86) \text{ b_eIy2.k_eIy1};$$

$$(\%o86) \quad 0$$

Calculating the average outcome of measuring ly for lalpha> and lbeta>

(%i87) b_a.Iy.k_a;

(%o87) 0

(%i88) b_b.Iy.k_b;

(%o88) 0

Calculating the average outcome of measuring ly for the eigenfunctions of lx

(%i89) b_eIx1.Iy.k_eIx1;

(%o89) 0

(%i90) b_eIx2.Iy.k_eIx2;

(%o90) 0

Calculating the average outcome of measuring ly for the ly eigenvectors

(%i91) b_eIy1.Iy.k_eIy1;

(%o91) $\frac{1}{2}$

(%i92) b_eIy2.Iy.k_eIy2;

(%o92) $-\frac{1}{2}$

The equations for calculating each of the average orthogonal magnetization components for a general wavefunction

(%i93) meanPsi(Ix,k_psi);

(%o93) $\frac{\overline{ca} \cdot cb + ca \cdot \overline{cb}}{2}$

(%i94) meanPsi(Iy,k_psi);

(%o94) $\frac{i \cdot ca \cdot \overline{cb} - i \cdot \overline{ca} \cdot cb}{2}$

(%i95) meanPsi(Iz,k_psi);

(%o95) $-\frac{cb \cdot \overline{cb} - ca \cdot \overline{ca}}{2}$

2 11.5 A possible source of confusion

Page 335

The example used in this subsection is one of the eigenfunctions of lx

(%i96) k_eIx1;

(%o96) $\begin{pmatrix} \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} \end{pmatrix}$

The result of applying the lz operator to this function is

(%i97) Iz.k_eIx1;

$$(\%o97) \quad \begin{pmatrix} \frac{1}{2^{\frac{3}{2}}} \\ -\frac{1}{2^{\frac{3}{2}}} \end{pmatrix}$$

Multiplying this with the bra of k_elx1

$$(\%i98) \quad \texttt{b_eIx1.}\%;$$

$$(\%o98) \quad 0$$

3 11.6 Application to the Stern-Gerlach experiments

Page 336

Taking particles with wavefunction k_a, from the z-filter and, again measuring Iz

$$(\%i99) \quad \texttt{k_a};$$

$$(\%o99) \quad \begin{pmatrix} 1 \\ 0 \end{pmatrix}$$

$$(\%i100) \quad \texttt{Iz.k_a};$$

$$(\%o100) \quad \begin{pmatrix} \frac{1}{2} \\ 0 \end{pmatrix}$$

$$(\%i101) \quad \texttt{b_a.Iz.k_a};$$

$$(\%o101) \quad \frac{1}{2}$$

Now measure Ix

$$(\%i102) \quad \texttt{Ix.k_a};$$

$$(\%o102) \quad \begin{pmatrix} 0 \\ \frac{1}{2} \end{pmatrix}$$

$$(\%i103) \quad \texttt{b_a.Ix.k_a};$$

$$(\%o103) \quad 0$$

Take the particles from the x-filter, with wavefunction k_elx1 and measure the Iz magnetization

$$(\%i104) \quad \texttt{k_eIx1};$$

$$(\%o104) \quad \begin{pmatrix} \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} \end{pmatrix}$$

$$(\%i105) \quad \texttt{Iz.k_eIx1};$$

$$(\%o105) \quad \begin{pmatrix} \frac{1}{2^{\frac{3}{2}}} \\ -\frac{1}{2^{\frac{3}{2}}} \end{pmatrix}$$

$$(\%i106) \quad \texttt{b_eIx1.Iz.k_eIx1};$$

$$(\%o106) \quad 0$$

4 11.7 Other angles

Other angles: Matrix representation of the operator for a Stern Gerlach filter oriented at angle phi from z-axis. In this case, the operator is a function of phi

(%i107) Iphi(phi):=(declare(phi,real),(1/2)*matrix([cos(phi),sin(phi)],[sin(phi),-cos(phi)]));

(%o107)
$$\text{Iphi}(\phi) := \left(\text{declare}(\phi, \text{real}), \frac{\begin{pmatrix} \cos(\phi) & \sin(\phi) \\ \sin(\phi) & -\cos(\phi) \end{pmatrix}}{2} \right)$$

(%i108) Iphi(phi);

(%o108)
$$\begin{pmatrix} \frac{\cos(\phi)}{2} & \frac{\sin(\phi)}{2} \\ \frac{\sin(\phi)}{2} & -\frac{\cos(\phi)}{2} \end{pmatrix}$$

(%i109) Iphi(phi).k_psi;

(%o109)
$$\begin{pmatrix} \frac{cb \cdot \sin(\phi)}{2} + \frac{ca \cdot \cos(\phi)}{2} \\ \frac{ca \cdot \sin(\phi)}{2} - \frac{cb \cdot \cos(\phi)}{2} \end{pmatrix}$$

Compare this operator, with phi=0, with Iz

(%i110) Iphi(0);

(%o110)
$$\begin{pmatrix} \frac{1}{2} & 0 \\ 0 & -\frac{1}{2} \end{pmatrix}$$

(%i111) Iz;

(%o111)
$$\begin{pmatrix} \frac{1}{2} & 0 \\ 0 & -\frac{1}{2} \end{pmatrix}$$

Compare Iphi(pi/2) with Ix

(%i112) Iphi(%pi/2);

(%o112)
$$\begin{pmatrix} 0 & \frac{1}{2} \\ \frac{1}{2} & 0 \end{pmatrix}$$

(%i113) Ix;

(%o113)
$$\begin{pmatrix} 0 & \frac{1}{2} \\ \frac{1}{2} & 0 \end{pmatrix}$$

(%i114) Iphi(phi).k_b;

(%o114)
$$\begin{pmatrix} \frac{\sin(\phi)}{2} \\ -\frac{\cos(\phi)}{2} \end{pmatrix}$$

Maxima can find the general form of the eigenvectors for this matrix, but has a hard time simplifying them to the forms shown in the text. But we can show that the ones given in the text satisfy the requirements of orthonormal eigenvectors.

(%i115) k_ePhi1(phi):=ket(cos(phi/2),sin(phi/2));

(%o115)
$$\text{k_ePhi1}(\phi) := \text{ket}\left(\cos\left(\frac{\phi}{2}\right), \sin\left(\frac{\phi}{2}\right)\right)$$

(%i116) k_ePhi1(phi);

```
(%o116)
```

$$\begin{pmatrix} \cos\left(\frac{\phi}{2}\right) \\ \sin\left(\frac{\phi}{2}\right) \end{pmatrix}$$

```
(%i117) b_ePhi1(phi):=bra(k_ePhi1(phi));
```

```
(%o117)  b_ePhi1 (phi) := bra (k_ePhi1 (phi))
```

```
(%i118) b_ePhi1(phi);
```

```
(%o118)  ⎛ cos ⎡ φ ⎤   sin ⎡ φ ⎤ ⎞
```

```
(%i119) b_ePhi1(phi).k_ePhi1(phi);
```

```
(%o119)  sin ⎡ φ ⎤2 + cos ⎡ φ ⎤2
```

```
(%i120) trigsimp(%);
```

```
(%o120)  1
```

```
(%i121) k_ePhi2(phi):=ket(-sin(phi/2),cos(phi/2));
```

```
(%o121)  k_ePhi2 (phi) := ket ⎛ −sin ⎡ φ ⎤ , cos ⎡ φ ⎤ ⎞
```

```
(%i122) k_ePhi2(phi);
```

```
(%o122)  ⎛ −sin ⎡ φ ⎤ ⎞
```

```
(%i123) b_ePhi2(phi):=bra(k_ePhi2(phi));
```

```
(%o123)  b_ePhi2 (phi) := bra (k_ePhi2 (phi))
```

```
(%i124) b_ePhi2(phi);
```

```
(%o124)  ⎛ −sin ⎡ φ ⎤   cos ⎡ φ ⎤ ⎞
```

```
(%i125) b_ePhi2(phi).k_ePhi2(phi);
```

```
(%o125)  sin ⎡ φ ⎤2 + cos ⎡ φ ⎤2
```

```
(%i126) trigsimp(%);
```

```
(%o126)  1
```

```
(%i127) b_ePhi1(phi).k_ePhi2(phi);
```

```
(%o127)  0
```

Demonstration that the proposed eigenfunctions really are eigenfunctions.

```
(%i128) Iphi(phi).k_ePhi1(phi);
```

```
(%o128)  ⎛ sin ⎡ φ ⎤ ·sin(φ)   cos ⎡ φ ⎤ ·cos(φ) ⎞
```

```
(%i129) trigrat(%);
```

$$(\%o129) \quad \begin{pmatrix} \frac{\cos\left(\frac{\phi}{2}\right)}{2} \\ \frac{\sin\left(\frac{\phi}{2}\right)}{2} \end{pmatrix}$$

```
(%i130) %/k_ePhi1(phi);
```

$$(\%o130) \quad \begin{pmatrix} \frac{1}{2} \\ \frac{1}{2} \end{pmatrix}$$

```
(%i131) Iphi(phi).k_ePhi2(phi);
```

$$(\%o131) \quad \begin{pmatrix} \frac{\cos\left(\frac{\phi}{2}\right)\cdot\sin(\phi)}{2} - \frac{\sin\left(\frac{\phi}{2}\right)\cdot\cos(\phi)}{2} \\ -\frac{\sin\left(\frac{\phi}{2}\right)\cdot\sin(\phi)}{2} - \frac{\cos\left(\frac{\phi}{2}\right)\cdot\cos(\phi)}{2} \end{pmatrix}$$

```
(%i132) trigrat(%);
```

$$(\%o132) \quad \begin{pmatrix} \frac{\sin\left(\frac{\phi}{2}\right)}{2} \\ -\frac{\cos\left(\frac{\phi}{2}\right)}{2} \end{pmatrix}$$

```
(%i133) %/k_ePhi2(phi);
```

$$(\%o133) \quad \begin{pmatrix} -\frac{1}{2} \\ -\frac{1}{2} \end{pmatrix}$$

Make phi=-pi/4 The operator is now

```
(%i134) Iphi(%pi/4);
```

$$(\%o134) \quad \begin{pmatrix} \frac{1}{2^{\frac{3}{2}}} & \frac{1}{2^{\frac{3}{2}}} \\ \frac{1}{2^{\frac{3}{2}}} & -\frac{1}{2^{\frac{3}{2}}} \end{pmatrix}$$

The eigenfunctions, with phi=-pi/4

```
(%i135) k_ePhi1(-%pi/4);
```

$$(\%o135) \quad \begin{pmatrix} \cos\left(\frac{\pi}{8}\right) \\ -\sin\left(\frac{\pi}{8}\right) \end{pmatrix}$$

```
(%i136) float(%);
```

$$(\%o136) \quad \begin{pmatrix} 0.9238795325112867 \\ -0.3826834323650898 \end{pmatrix}$$

```
(%i137) k_ePhi2(-%pi/4);
```

$$(\%o137) \quad \begin{pmatrix} \sin\left(\frac{\pi}{8}\right) \\ \cos\left(\frac{\pi}{8}\right) \end{pmatrix}$$

```
(%i138) float(%);
```

$$(\%o138) \quad \begin{pmatrix} 0.3826834323650898 \\ 0.9238795325112867 \end{pmatrix}$$

Applying the phi=-pi/4 operator to lalpha> to find the average outcome.


```
(%i139) b_a.Iphi(-%pi/4).k_a;
```

```
(%o139) 1/2^(3/2)
```

```
(%i140) float(%);
```

```
(%o140) 0.3535533905932737
```

Calculating the probabilities

```
(%i141) abs(b_ePhi1(-%pi/4).k_a)^2;
```

```
(%o141) cos(pi/8)^2
```

```
(%i142) float(%);
```

```
(%o142) 0.8535533905932737
```

```
(%i143) abs(b_ePhi2(-%pi/4).k_a)^2;
```

```
(%o143) sin(pi/8)^2
```

```
(%i144) float(%);
```

```
(%o144) 0.1464466094067262
```

5 11.8 The shift operators I_+ and I_-

Page 339

```
(%i145) Ix+%i*Iy;
```

```
(%o145) (0 1)
         (0 0)
```

```
(%i146) Ix-%i*Iy;
```

```
(%o146) (0 0)
         (1 0)
```

These are defined in 1spinLib.mac

```
(%i147) Iplus;
```

```
(%o147) (0 1)
         (0 0)
```

```
(%i148) Iminus;
```

```
(%o148) (0 0)
         (1 0)
```

Shift operators applied to a general wavefunction

```
(%i149) Iplus.k_psi;
```

```
(%o149) (cb)
         (0)
```

```
(%i150) Iminus.k_psi;
```

(%o150)

$$\begin{pmatrix} 0 \\ ca \end{pmatrix}$$

Shift operators applied to the Iz eigenfunctions

(%i151) Iplus.k_a;

(%o151)

$$\begin{pmatrix} 0 \\ 0 \end{pmatrix}$$

(%i152) Iplus.k_b;

(%o152)

$$\begin{pmatrix} 1 \\ 0 \end{pmatrix}$$

(%i153) Iminus.k_a;

(%o153)

$$\begin{pmatrix} 0 \\ 1 \end{pmatrix}$$

(%i154) Iminus.k_b;

(%o154)

$$\begin{pmatrix} 0 \\ 0 \end{pmatrix}$$